



IIC2343 - Arquitectura de Computadores (II/2025)

Ayudantía 9

Ayudantes: Daniela Ríos (danielaarp@uc.cl), Alberto Maturana (alberto.maturana@uc.cl), Mario Rojas (mario.denzel@estudiante.uc.cl)

Pregunta 1: DCCadabra - Análisis del grimorio

La **Escuela de Hechiceros del DCC** se prepara para una gran batalla final contra las fuerzas del Caos Binario. Cada aprendiz de mago del Departamento debe fortalecer los sistemas mágicos de la escuela y optimizar el uso del maná, la fuente de poder que alimenta todos los hechizos.

Para ello, el Consejo de Hechicería Computacional te ha encomendado desarrollar un programa en **RISC-V** que analice el *grimorio de energía mágica*, representado como un arreglo en memoria. Este grimorio contiene los costos de maná de cada hechizo. Los más poderosos, sin embargo, superan un umbral de energía permitido (**LIMITE_MAGICO**) y están estrictamente prohibidos.

Implemente un programa que:

1. Recorra el arreglo **mana**, que contiene los costos de maná de cada hechizo.
2. Compare cada valor con la constante **LIMITE_MAGICO**.
3. Si el valor es menor o igual al límite, sume su valor al acumulador **energia_total**.
4. Si el valor supera el límite, incremente el contador **hechizos_prohibidos**.
5. Guarde ambos resultados en memoria y haga que el programa los muestre por consola mediante **ecall**.

Puede usar la siguiente estructura base:

```
.data
mensaje_energia: .string "Energia_total:␣"
mensaje_prohibido: .string "Hechizos_prohibidos:␣"
enter: .string "\n"
mana: .word 25, 60, 40, 15, 75, 50, 20, 10
LIMITE_MAGICO: .word 50
energia_total: .word 0
hechizos_prohibidos: .word 0
largo: .word 8

.text
main:
    #Completa el codigo :D

end:
    #Aqui tambien debes completar
    addi a7, zero, 4 #Codigo para imprimir strings
    la a0, mensaje_energia #Cargamos la direccion del string como argumento
    ecall #Imprimimos

    addi a7, zero, 4
    la a0, enter
    ecall

    addi a7, zero, 4
    la a0, mensaje_prohibido
    ecall

    addi a7, zero, 10
    ecall
```

Solución: A continuación, un código que cumple lo pedido:

```
.data
mensaje_energia: .string "Energia_total:_"
mensaje_prohibido: .string "Hechizos_prohibidos:_"
enter: .string "\n"
mana: .word 25, 60, 40, 15, 75, 50, 20, 10
LIMITE_MAGICO: .word 50
energia_total: .word 0
hechizos_prohibidos: .word 0
largo: .word 8

.text
main:
    la t0, mana           # Dirección base del array
    lw t1, largo          # Largo = 8
    lw t2, LIMITE_MAGICO
    addi t3, zero, 0      # Suma total de energía válida
    addi t4, zero, 0      # Contador de hechizos prohibidos

loop:
    beq t1, zero, end     # Si ya recorrimos todo, salimos
    lw t5, 0(t0)          # Cargar mana[i]
    blt t2, t5, prohibido # Si límite < mana[i] , va a prohibido
    add t3, t3, t5         # Sumar energía válida
    jal zero, siguiente

prohibido:
    addi t4, t4, 1        # Contar hechizo prohibido

siguiente:
    addi t0, t0, 4        # Avanzar a siguiente elemento
    addi t1, t1, -1
    jal zero, loop

end:
    la t6, energia_total
    sw t3, 0(t6)
    la t6, hechizos_prohibidos
    sw t4, 0(t6)

    addi a7, zero, 4
    la a0, mensaje_energia
    ecall
    addi a7, zero, 1
    lw a0, energia_total
    ecall
    addi a7, zero, 4
    la a0, enter
    ecall
    addi a7, zero, 4
    la a0, mensaje_prohibido
    ecall
    addi a7, zero, 1
    lw a0, hechizos_prohibidos
    ecall
    addi a7, zero, 10
    ecall
```

Pregunta 2: DDCadabra - La revisión de las artes arcanas

Tras analizar el grimorio, el **Consejo de Hechicería del DCC** exige que todos los aprendices aseguren la pureza de su código conforme a las normas ancestrales del ensamblador: la **Convención de llamadas de RISC-V**.

Tus superiores observan atentamente tu progreso y esperan que tus conjuros arcanos mantengan el orden y la disciplina mágica que rige toda la programación del Consejo.

Instrucciones:

1. Analiza tu código de la Parte 1 e indica si cumple o no la convención de llamadas.
2. Justifica tu respuesta identificando los registros que no se preservan correctamente o los argumentos mal utilizados.
3. En caso de no cumplirla, modifica tu código para ajustarse a la convención de llamadas de RISC-V.
4. Describe brevemente los cambios realizados.

Recordatorio de la convención

- a0-a7: argumentos y valores de retorno (caller-saved).
- t0-t6: temporales (caller-saved).
- s0-s11: registros preservados (callee-saved).
- sp: puntero de pila.
- ra: dirección de retorno.

Solución: A continuación, un código que cumple lo pedido:

```
.data
mensaje_energia: .string "Energia_total:_"
mensaje_prohibido: .string "Hechizos_prohibidos:_"
enter: .string "\n"
mana: .word 25, 60, 40, 15, 75, 50, 20, 10
LIMITE_MAGICO: .word 50
energia_total: .word 0
hechizos_prohibidos: .word 0
largo: .word 8

.text
main:
    la t0, mana
    lw t1, largo
    lw s0, LIMITE_MAGICO # Al ser una constante debe ser igual aun en subrutinas
    addi s1, zero, 0     # El resultado y el contador son valores que queremos que se mantengan siempre
    addi s2, zero, 0

loop:
    beq t1, zero, end
    lw t2, 0(t0)
    blt s0, t2, prohibido
```

```

    add s1, s1, t2
    jal zero, siguiente

prohibido:
    addi s2, s2, 1

siguiente:
    addi t0, t0, 4
    addi t1, t1, -1
    jal zero, loop

end:
    la t6, energia_total
    sw s1, 0(t6)
    la t6, hechizos_prohibidos
    sw s2, 0(t6)

    addi a7, zero, 4
    la a0, mensaje_energia
    ecall
    addi a7, zero, 1
    lw a0, energia_total
    ecall
    addi a7, zero, 4
    la a0, enter
    ecall
    addi a7, zero, 4
    la a0, mensaje_prohibido
    ecall
    addi a7, zero, 1
    lw a0, hechizos_prohibidos
    ecall
    addi a7, zero, 10
    ecall

```

Pregunta 3: DCCadabra - La maldición de los magos enemigos

Cuando creías haber dominado las artes arcanas, una orden oscura de magos enemigos lanza un ataque contra los grimorios del gremio, corrompiendo sus hechizos y transformando varios valores en números negativos. Tus superiores continúan observándote desde el Consejo, y esperan que, incluso en medio del caos, mantengas el orden del código y sigas aplicando correctamente la **Convención de llamadas de RISC-V** en tus conjuros.

Para restaurar el equilibrio mágico, implementa una **subrutina** llamada `purificar_hechizo`, que reciba un valor (por ejemplo, en `a0`) y:

- Si el valor es negativo, lo convierta a su equivalente positivo.
- Si el valor es positivo, lo mantenga igual.
- Retorne el valor purificado en el mismo registro `a0`.

```
.data
mana: .word -25, 60, -40, 15, 75, -50, 20, -10 #Grimorio corrupto
```

Solución: A continuación, un código que cumple lo pedido:

```
.data
mensaje_energia: .string "Energia_total:_"
mensaje_prohibido: .string "Hechizos_prohibidos:_"
enter: .string "\n"
mana: .word -25, 60, -40, 15, 75, -50, 20, -10
limite_magico: .word 50
energia_total: .word 0
hechizos_prohibidos: .word 0
largo: .word 8

.text
main:
    la t0, mana
    lw t1, largo
    lw s0, limite_magico
    addi s1, zero, 0           # energia total
    addi s2, zero, 0           # hechizos prohibidos

loop:
    beq t1, zero, end
    lw a0, 0(t0)

    # Caller guarda los registros antes de llamar
    addi sp, sp, -16
    sw ra, 0(sp)
    sw a0, 4(sp)
    sw t0, 8(sp)
    sw t1, 12(sp)

    jal ra, purificar_hechizo # Llamada a función

    add t5, zero, a0 #Guardamos el argumento antes de restaurar

    # Caller restaura los registros antes de llamar
    lw ra, 0(sp)
    lw a0, 4(sp)
    lw t0, 8(sp)
    lw t1, 12(sp)
    addi sp, sp, 16

    blt s0, t5, prohibido
    add s1, s1, t5
    j siguiente

prohibido:
    addi s2, s2, 1

siguiente:
    addi t0, t0, 4
    addi t1, t1, -1
    j loop

end:
    la t6, energia_total
    sw s1, 0(t6)
    la t6, hechizos_prohibidos
    sw s2, 0(t6)

    addi a7, zero, 4
    la a0, mensaje_energia
    ecall
    addi a7, zero, 1
    lw a0, energia_total
    ecall
```

```
    addi a7, zero, 4
    la a0, enter
    ecall
    addi a7, zero, 4
    la a0, mensaje_prohibido
    ecall
    addi a7, zero, 1
    lw a0, hechizos_prohibidos
    ecall
    addi a7, zero, 10
    ecall

# --- Subrutina ---
purificar_hechizo:
    blt a0, zero, negativo
    jalr zero, 0(ra)

negativo:
    addi t0, zero, -1
    mul a0, a0, t0
    jalr zero, 0(ra)
```


Feedback ayudantía

Escanee el QR para entregar feedback sobre la ayudantía.

